適用於全球預報模式極方位投影系統之客觀分析研究

林松錦＊　喬鳳倫＊＊　方力脩＊＊

＊國立中央大學大氣物理系
＊＊中央氣象局

摘　要

中央氣象局現行全球數值天氣預報系統在模式中之運算係在經緯度網格點上進行，但在顯示預報結果時却以半球的極方位（ polar stereographic ） 投影地圖較方便於繪製天氣圖。本研究之主要目的為配合全球數值天氣預報系統的設計，發展一套由經緯度網格系統轉換至極方位投影直角網格系統之應用程式軟體。

一、前言

在大氣科學研究的領域裏，不論是作診斷分析或是數值模式預報，所採用的網格系統大致可分成兩類，一類是直接採用經緯度網格系統，另一類是地圖投影網格系統。經緯度網格系統對實際距離的處理不涉及地圖投影的問題，在分析計算及模式設計上較為方便。在作全球或半球的診斷分析或是數值模式運算時，通常是採用經緯度網格系統，而在顯示分析結果時却又以使用半球投影的極方位投影網格系統（ polar steorographic projection grid system ）較為方便。中央氣象局的全球預報系統在模式運算時即是採用經緯度網格系統，因此在顯示模式輸出時還需將預報結果轉換至極方位投影網格以便繪製天氣圖。本研究之主要目的為配合全球預報系統之設計，發展一套座標轉換系統之程式軟體。

二、極方位投影網格系統

極方位投影網格通常使用於半球性之分析。地圖投影曲面為一垂直於地球轉軸且切割於某一標準緯度之平面。在標準緯度圈上，投影曲面上之距離與地球表面上之實際距離相等。由投影產生之地圖具正形（ conformal ）特性，在投影面上經緯度線呈以極點為中心之輻射線及同心圓。極方位投影地圖之影像比例（ image scale ）為緯度的函數，可以下式表示

$$\sigma(\phi)=\frac{1+H\sin\phi_0}{1+H\sin\phi} \quad \cdots\cdots\cdots\cdots\cdots \quad (1)$$

其中 ϕ 為緯度，ϕ_0 為標準緯度而H在北半球為1在南半球為－1。全球預報系統使用之極方位投影地圖，其標準緯度分別訂在北緯 60 度及南緯60度。有關地圖投影之數學導演可參閱Saucier（ 1955 ），Jenne（ 1970 ），Hoke等（ 1980）及曾（ 1984 ）。



圖 1 ：北半球極方位投影直角座標(x,y)及網格系統(I,J)。

圖1及圖2分別是全球預報系統所使用在極方位投影地圖上南北半球之直角座標（x，y）及網格分佈（I，J）。直角座標之原點分別設在南、北極，而其網格位置都是（32，32）。網格點之間的距離爲381公里，以這種網格間距而涵蓋整個半球的網格點應是65×65。但爲配合美國海軍FNOC（Fleet Numerical Oceangraphic Center）所儲存的氣候資料，全球預報系統所採用的網格點是63×63，亦即和地圖四邊相切的緯度約在0.3765度。直角座標和經緯度之間可經由下列兩式轉換

$$x = \sigma a \cos\phi \cos(\lambda - \lambda_0) = \gamma \cos\theta \qquad (2)$$

$$y = \sigma a \cos\phi \sin(\lambda - \lambda_0) = \gamma \sin\theta \qquad (3)$$

兩式中

$$\gamma = \sigma a \cos\phi \qquad (4)$$

$$\theta = \lambda - \lambda_0 \qquad (5)$$

a爲地球半徑，λ 爲經度，λ_0 爲平行於x軸的參考經度。圖1及圖2中南北半球地圖之參考經度均



圖2：南半球極方位投影直角座標(x,y)及網格系統(I,J)。

設在西經150度。

由於在經緯度網格及地圖投影直角座標上風速分量定義的方向不同，因此風速分量除需做位置內插外，還需做方向的轉換。參考圖3假設U及V分別代表在經緯度座標上向東及向北之風速分量，而 U_g 及 V_g 分別代表在地圖投影直角座標上沿x及y方向之風速分量，則兩種風速分量之間的相互轉換可以下列(6)～(9)式表示

$$U_g = U\cos\alpha - V\sin\alpha \quad \cdots\cdots\cdots\cdots\cdots\cdots \quad (6)$$

$$V_g = U\sin\alpha + V\cos\alpha \quad \cdots\cdots\cdots\cdots\cdots\cdots \quad (7)$$

$$U = U_g\cos\alpha + V_g\sin\alpha \quad \cdots\cdots\cdots\cdots\cdots \quad (8)$$

$$V = -U_g\sin\alpha + V_g\cos\alpha \quad \cdots\cdots\cdots \quad (9)$$

式中之 α 爲兩座標之間的夾角，而圖中之夾角 γ 爲直角網格點之位置向量和x軸之間的夾角，由圖中可見 $\alpha+\gamma=\pi/2$。(6)～(9)式並不僅限使用於極方向投影地圖之直角座標，若將（x，y）直角座標的原點固定在兩極，則亦可應用於藍伯特投影或麥卡脫投影地圖之直角座標。



圖3：在經緯度座標及直角座標上風速分量之關係。

三、程式設計與說明

在中央氣象局數值天氣預報系統中，所有有關的網格系統均由一特有之DMS（Data Management System，CDC（1984））file'GRIDDEF'來定義。完整的GRIDDEF file應包含有六個record分別定義全球經緯度、北半球極方位、南半球極方位、區域，中尺度及颱風路徑等六個網格系統。每一個record由20個整數組成，每一個整數儲存資料的定義請參閱附錄一或Liou（1987）。附錄二爲產生GRIDDEF file的程式，程式中颱風路徑網格系統因尚未確定並未列入。

附錄三中的程式LTLNIJ爲實際作網格座標轉換及資料內插的主程式。在全球預報模式中資料輸出是以2.5度經緯度網格（73×144）儲存，和經內插後之極方位網格資料一起儲存在一DMS file'MASCWB'內。MASCWB file中每一氣象變數是以一DMS record key來辨認，一組record key由24個字符組成，其定義可參閱附錄四。主程式中做資料內插時，若極方位網格點位於經緯度網格由外向內算第二層之內時採16點Bessel內插；餘則採4點線性內插。4點及16點內插可分別參照圖4及圖5以下列二式表示

$$\phi_A = \Delta i\phi_2 + (1-\Delta i)\phi_1$$
$$\phi_B = \Delta i\phi_4 + (1-\Delta i)\phi_3 \quad \cdots\cdots\cdots (10)$$
$$\phi_P = \Delta j\phi_A + (1-\Delta j)\phi_B$$

$$\phi_A = \phi_2 + \Delta i\,[\,(\phi_3-\phi_2) + (\Delta i-1)(\phi_1-\phi_2-\phi_3+\phi_4)/4\,]$$
$$\phi_B = \phi_6 + \Delta i\,[\,(\phi_7-\phi_6) + (\Delta i-1)(\phi_5-\phi_6-\phi_7+\phi_8)/4\,]$$
$$\phi_C = \phi_{10} + \Delta i\,[\,(\phi_{11}-\phi_{10}) + (\Delta i-1)(\phi_9-\phi_{10}-\phi_{11}+\phi_{12})/4\,]$$
$$\phi_D = \phi_{14} + \Delta i\,[\,(\phi_{15}-\phi_{14}) + (\Delta i-1)(\phi_{13}-\phi_{14}-\phi_{15}+\phi_{16})/4\,]$$
$$\phi_P = \phi_C + \Delta j\,[\,(\phi_B-\phi_C) + (\Delta j-1)(\phi_A-\phi_B-\phi_C+\phi_D)/4\,] \cdots (11)$$

在資料內插過程中，爲避免因內插產生小擾動，每一變數經內插後均經下式25點修勻，

$$\begin{aligned}\overline{\phi}_{i,j} = {} & 0.279372\,\phi_{i,j} \\ & + 0.171943\,(\phi_{i,j+1}+\phi_{i,j-1}+\phi_{i+1,j}+\phi_{i-1,j}) \\ & - 0.006918\,(\phi_{i,j+2}+\phi_{i,j-2}+\phi_{i+2,j}+\phi_{i-2,j}) \\ & + 0.077458\,(\phi_{i+1,j+1}+\phi_{i+1,j-1}+\phi_{i-1,j+1}+\phi_{i-1,j-1}) \\ & - 0.024693\,(\phi_{i+2,j+1}+\phi_{i+2,j-1}+\phi_{i+1,j+2}+\phi_{i+1,j-2}+\phi_{i-1,j+2}+\phi_{i-1,j-2}+\phi_{i-2,j+1}+\phi_{i-2,j-1}) \\ & - 0.012940\,(\phi_{i+2,j+2}+\phi_{i+2,j-2}+\phi_{i-2,j+2}+\phi_{i-2,j-2}) \quad \cdots\cdots (12)\end{aligned}$$

$\overline{\phi}_{i,j}$代表經修勻後的值。



圖４ ：４點線性內插網格點分佈，Ｐ點代表欲內插之直角座標網格點。



圖５ ：16點Bessel內插網格點分佈，Ｐ點代表欲內插之直角座標網格點。

參考文獻

曾忠一，1984：動力預報的基本方程。122頁。

Control Data Corporation , 1986 : Central Weather Bureau data management system, user manual. 26pp.

Hoke , J.E., J.L. Hayes and L.G. Renninger , 1981 : Map projections and grid systems for meteorological applications. U.S. Air Force Global Weather Center , AFGWC/TN-79/003, 86pp.

Jenne , R.L. , 1970 : The NMC Octagonal grid. NCAR Memo.

Liou , C. -S ., 1987 : Lectures in description of CWB system control procedures and programs. I. CWB system control overview. NWP short course 1987-5, 78pp.

Saucier, W. J., 1955 : Principles of Meteorological Analysis. University of Chicago Press, Chicago, IL, 438pp.

附錄一

```
DMS file name:    GRIDDEF
Residence:        FE CYBER-840
Purpose:          A file to define grid informations for NWP models
Format:           20 integer words defined as below
Key defination:   24 characters defined as the following,
                  1 - 6: blanks,
                  7 - 8: grid ID (GA-144x73 global lat/lon grid,
                                  NA-63x63 northern hemispheric
                                     grid on polar stereographic
                                     projection map,
                                  SA-63x63 southern hemispheric
                                     grid on polar stereographic
                                     projection map,
                                  RA-111x81 regional grid on Lambert
                                     conformal projection map,
                                  MA-121x106 mesoscale grid on
                                     Lambert conformal projection map,
                                  TA-131x111 typhoon grid on Lambert
                                     conformal projection map),
                  9 -16: blanks
                     17: record type 'I'
                 18 -24: record length in units of record type '0000020'

Word  1: projection type - 4HLTLN (lat/lon grid)
                           4HNHPS (northern hemi. polar stereographic)
                           4HSHPS (southern hemi. polar stereographic)
                           4HMERC (Mercator)
                           4HLAMB (Lambert conformal)
      2: true latitude 1 (in units of 1/1000 degree; +N, -S)
      3: true latitude 2 (in units of 1/1000 degree; -999999 means no
                          second true latitude)
      4: for projection type NHPS, SHPS and LAMB: central longitude
                 which is parallel to y-axis, +E, -W; in 1/1000 degree
         for projection type LTLN and MERC: rotating angle (counter-
                 clockwise) in 1/1000 degree
      5: grid x-dimension, M
      6: grid y-dimension, N
      7: latitude of point (1,1), in 1/1000 degree; +N, -S
      8: longitude of point (1,1), in 1/1000 degree; +E, -W
      9: x-grid size; for LTLN grid in 1/1000 degree, for other types
                      in meters at the true latitude
     10: y-grid size; for LTLN grid in 1/1000 degree, for other types
                      in meters at the true latitude
     11: parent grid ID (for zoom plot otherwise same as grid ID
                         in the key)
     12: latitude of point (M,N), in 1/1000 degree; +N, -S
     13: longitude of point (M,N), in 1/1000 degree; +E, -W
     14: x-coordinate of the pole point, in units of 1/1000 grid scale
     15: y-coordinate of the pole point, in units of 1/1000 grid scale
     16: map distance from the pole point to point (1,1), in meters
     17: mapping constant x 1000 for Lambert conformal projection
     18: number of vertical levels
     19: reserved
     20: reserved

* NOTE: all non-applicable parameters are indicated by -999999
```

附錄二

```
      PROGRAM CGRID(INPUT,OUTPUT)
C
C     PROGRAM TO CREATE GRID DEFINITION FILE 'GRIDDEF'
C
C     NO INPUT DATA REQUIRED
C
C     REFERENCE:
C
C     LIOU, C.-S., 1987: LECTURES IN DESCRIPTION OF CWB SYSTEM CONTROL
C          PROCEDURES AND PROGRAMS. I. CWB SYSTEM CONTROL OVERVIEW.
C          NWP SHORT COURSE 1987-5, 74-75.
C
C     ENTRY POINTS:
C
C          SYSTEM LIBRARY:
C
C               DMSCLS    DMSMSG    DMSOPN    DMSPUT    EXIT
C
      CHARACTER*7  DFILE
      CHARACTER*24 KEYGA, KEYNA, KEYSA, KEYRA, KEYMA
      DIMENSION IGGA(20), IGNA(20), IGSA(20), IGRA(20), IGMA(20)
      DATA IGGA, IGNA, IGSA, IGRA, IGMA
     +     /20*-999999, 20*-999999, 20*-999999, 20*-999999, 20*-999999/
      DFILE='GRIDDEF'
      CALL DMSMSG('ERR')
      CALL DMSOPN(DFILE,ISTAT)
      IF(ISTAT .NE. 0) THEN
         PRINT 5000, DFILE
 5000    FORMAT(1X,'DMS FILE "',A7,'" OPEN ERROR')
         CALL EXIT
      END IF
C
C     LATITUDE/LONGITUDE GRID FOR GLOBAL FORECAST MODEL
C
      IGGA(1) = 4HLTLN
      IGGA(4) = 0
      IGGA(5) = 144
      IGGA(6) = 73
      IGGA(7) = -90000
      IGGA(8) = 0
      IGGA(9) = 2500
      IGGA(10) = 2500
      IGGA(11) = 2HGA
      IGGA(12) = 90000
      IGGA(13) = -2500
      WRITE (KEYGA,1000) IGGA(11)
 1000 FORMAT(6X,A2,8X,8HI0000020)
      CALL DMSPUT(DFILE,KEYGA,IGGA,ISTAT)
      IF(ISTAT .NE. 0) THEN
         PRINT 5050, KEYGA
 5050    FORMAT(1X,'DMS RECORD "',A24,'" WRITE FAILURE')
      END IF
C
C     POLAR STEREOGRAPHIC, RECTANGULAR GRID FOR NORTHERN HEMISPHERE
C
```

```
      IGNA(1) = 4HNHPS
      IGNA(2) = 60000
      IGNA(4) = 120000
      IGNA(5) = 63
      IGNA(6) = 63
      IGNA(7) = -19116
      IGNA(8) = 75000
      IGNA(9) = 381000
      IGNA(10) = 381000
      IGNA(11) = 2HNA
      IGNA(12) = -19116
      IGNA(13) = -105000
      IGNA(14) = 32000
      IGNA(15) = 32000
      IGNA(16) = 16703280
      IGNA(17) = 1000
      WRITE (KEYNA,1000) IGNA(11)
      CALL DMSPUT(DFILE,KEYNA,IGNA,ISTAT)
      IF(ISTAT .NE. 0) PRINT 5050, KEYNA
C
C     POLAR STEREOGRAPHIC, RECTANGULAR GRID FOR SOUTHERN HEMISPHERE
C
      IGSA(1) = 4HSHPS
      IGSA(2) = -60000
      IGSA(4) = 120000
      IGSA(5) = 63
      IGSA(6) = 63
      IGSA(7) = 19116
      IGSA(8) = 75000
      IGSA(9) = 381000
      IGSA(10) = 381000
      IGSA(11) = 2HSA
      IGSA(12) = 19116
      IGSA(13) = -105000
      IGSA(14) = 32000
      IGSA(15) = 32000
      IGSA(16) = 16703280
      IGSA(17) = 1000
      WRITE (KEYSA,1000) IGSA(11)
      CALL DMSPUT(DFILE,KEYSA,IGSA,ISTAT)
      IF(ISTAT .NE. 0) PRINT 5050, KEYSA
C
C     LAMBERT CONFORMAL, RECTANGULAR GRID FOR REGIONAL FORECAST MODEL
C
      IGRA(1) = 4HLAMB
      IGRA(2) = 10000
      IGRA(3) = 40000
      IGRA(4) = 110000
      IGRA(5) = 111
      IGRA(6) = 81
      IGRA(7) = -8443
      IGRA(8) =  70045
      IGRA(9) = 90000
      IGRA(10) = 90000
      IGRA(11) = 2HRA
      IGRA(12) = 50652
      IGRA(13) = 177976
      IGRA(14) = 56000
```

```
      IGRA(15) = 179849
      IGRA(16) = 16849240
C
C     CONSTANT OF THE CONE = 0.4275969
C
      IGRA(17) = 428
      WRITE (KEYRA,1000) IGRA(11)
      CALL DMSPUT(DFILE,KEYRA,IGRA,ISTAT)
      IF(ISTAT .NE. 0) PRINT 5050, KEYRA
C
C     LAMBERT CONFORMAL, RECTANGULAR GRID FOR MESOSCALE FORECAST MODEL
C
      IGMA(1) = 4HLAMB
      IGMA(2) = 10000
      IGMA(3) = 40000
      IGMA(4) = 110000
      IGMA(5) = 121
      IGMA(6) = 106
      IGMA(7) = 12258
      IGMA(8) = 101634
      IGMA(9) = 30000
      IGMA(10) = 30000
      IGMA(11) = 2HMA
      IGMA(12) = 38636
      IGMA(13) = 141573
      IGMA(14) = 31000
      IGMA(15) = 480547
      IGMA(16) = 14423830
C
C     CONSTANT OF THE CONE = 0.4275969
C
      IGMA(17) = 428
      WRITE (KEYMA,1000) IGMA(11)
      CALL DMSPUT(DFILE,KEYMA,IGMA,ISTAT)
      IF(ISTAT .NE. 0) PRINT 5050, KEYMA
      CALL DMSCLS(DFILE,ISTAT)
      STOP 'PROGRAM "CGRID" DONE !!!'
      END
```

附錄三

```
      PROGRAM LTLNIJ(INPUT,OUTPUT,NWPDATA,TAPE4=NWPDATA)
C
C     PROGRAM TO PERFORM INTERPOLATION OF DATA FROM LAT/LON GRID OF
C     CWB NWP MODEL TO 63*63 RECTANGULAR POLAR STEREOGRAPHIC GRID AND
C     COORDINATES TRANSFORMATION FROM TRUE E-W/S-N COORDINATES TO
C     RECTANGULAR POLAR STEREOGRAPHIC GRID COORDINATES FOR WIND
C     COMPONENTS.
C
C     REQUIRED DMS FILE: MASCWB
C
C     INPUT SECTION:
C
C          INPUT DATA FILE: TAPE8 = NWPDATA
C
C               READ (8,1000,END=9999) KEYA, KEYB
C          1000 FORMAT(A24,1X,A24)
C
C          WHERE
C
C          KEYA.....DMS RECORD KEY FOR THE FIRST DATA FIELD
C          KEYB.....DMS RECORD KEY FOR THE SECOND DATA FIELD
C
C          FOR SINGLE FIELD CONTOURING, LEAVE KEYB TO BE BLANK
C
C          FOR OVER-CONTOURING OF TWO FIELDS (SUCH AS HEIGHT AND
C          TEMPERATURE), THE FIRST DATA FIELD WILL BE DRAWN IN SOLID
C          LINES AND THE SECOND DATA FIELD WILL BE DRAWN IN DASHED
C          LINES
C
C          FOR STREAMLINES, WIND VECTORS AND WIND BARBS, DATA FIELDS
C          REQUIRED ARE (U,V) WIND COMPONENTS.  HOWEVER, ONLY ONE DMS
C          RECORD KEY IS REQUIRED WITH THE SECOND AND THIRD CHARACTERS
C          REPLACED BY TWO SPECIAL DIGITS. (SEE BELOW)
C
C          FOR PLOTTING WIND FIELDS (STREAMLINES, WIND VECTORS AND WIND
C          BARBS) AND OTHER SPECIAL FEATURES, OPERATIONS ARE IDENTIFIED
C          BY THE SECOND AND THIRD CHARACTERS OF A DMS RECORD KEY.
C          VALID SPECIAL FEATURES ARE:
C
C               '91'.....STREAMLINE
C               '92'.....WIND VECTOR
C               '93'.....WIND BARB
C               '94'.....WIND VECTOR WITH ISOTACH
C               '95'.....WIND BARB WITH GEOPOTENTIAL HEIGHT
C
C          FOR EXAMPLE, DMS RECORD KEY FOR 500 MB STREAMLINES MAY BE
C          'F91000NA87120212F0003969'.
C
C     ENTRY POINTS:
C
C          PROGRAM PROVIDED:
C
C               FILT25    PSGRID    UVCVRT    WINDCV
C
C          SYSTEM LIBRARY:
```

```
C
C                DMSCLS     DMSGET     DMSMSG     DMSOPN     DMSPUT
C                EXIT
C
      PARAMETER (ILN=73,JLN=144,IPS=63,JPS=63)
      DIMENSION A(ILN,JLN), B(IPS,JPS), C(IPS,JPS)
      CHARACTER*7  DFILE
      CHARACTER*24 KEYA, KEYB
      CHARACTER*24 KEYG, KEYP
      COMMON /DMS/ DFILE, KEYG, KEYP
      COMMON /ANG/ PI4, PI, DEGRAD
      DATA IJLTN,OI,OJ,DX,DY /0,32.,32.,381.,381./
      DFILE='MASCWB '
      KEYG='      GA        H0010512'
      PI4=ATAN(1.)
      PI=4.*PI4
      DEGRAD=PI/180.
      CALL DMSMSG('ERR')
      CALL DMSOPN(DFILE,ISTATD)
      IF(ISTATD .NE. 0) THEN
         PRINT 5000, DFILE
 5000    FORMAT(1X,'DMS FILE "',A7,'" OPEN ERROR')
         CALL EXIT
      END IF
C
C     READ IN DMS RECORD KEY FOR OUTPUT POLAR STEREOGRAPHIC GRID DATA
C
 1111 READ (4,1000,END=9999) KEYA, KEYB
 1000 FORMAT(A24,1X,A24)
C
C     CHECK FOR SPECIAL KEYS FOR WIND FIELDS
C
      IF(KEYA(2:2) .EQ. '9') THEN
         KEYA(2:3)='20'
         KEYB(1:24)=KEYA(1:24)
         KEYB(2:3)='21'
      END IF
      IAB=1
 2222 IF(IAB .EQ. 1) THEN
         KEYP(1:24)=KEYA(1:24)
      ELSE
         KEYP(1:24)=KEYB(1:24)
      END IF
      IF(KEYP(1:6) .EQ. '      ') GO TO 1111
C
C     SKIP NON-HEMISPHERIC DATA FIELDS
C
      IF(KEYP(7:7) .NE. 'N' .AND. KEYP(7:7) .NE. 'S') THEN
         PRINT 5060, KEYP
 5060    FORMAT(1X,'SKIP NON-HEMISPHERIC DATA FIELD "',A24,'"')
         GO TO 1111
      END IF
C
      IF(KEYP(7:7) .EQ. 'N') THEN
         TLAT=60.
         AXSLON=-150.
      ELSE
         TLAT=-60.
```

```
         AXSLON=-150.
      END IF
C
C     KEYG.....DMS RECORD KEY FOR INPUT LAT/LON GRID DATA FIELD
C
      KEYG(1:6) =KEYP(1:6)
      KEYG(9:16)=KEYP(9:16)
      CALL DMSGET(DFILE,KEYG,A,ISTATG)
      IF(ISTATG .NE. 0) THEN
         PRINT 5020, KEYG
 5020    FORMAT(1X,'DATA FIELD "',A24,'" NOT AVAILABLE')
         GO TO 1111
      END IF
      DO 100 I=1,IPS
      AI=FLOAT(I)
      DO 100 J=1,JPS
      AJ=FLOAT(J)
      CALL PSGRID(AI,AJ,ALAT,ALON,IJLTN,OI,OJ,TLAT,DX,DY,AXSLON)
      XI=(90.-ALAT)/2.5+1.
      IN=IFIX(XI)
      IF(IN .GE. ILN) IN=IN-1
      IF(ALON .LT. 60.) ALON=ALON+360.
      YJ=(ALON-60.)/2.5+1.
      JN=IFIX(YJ)
      IF(JN .GE. JLN) JN=JN-1
      DI=XI-FLOAT(IN)
      DJ=YJ-FLOAT(JN)
      IF(IN .LT. 2 .OR. JN .LT. 2) GO TO 200
      IF((ILN-IN) .LT. 2 .OR. (JLN-JN) .LT. 2) GO TO 200
C
C     16-POINT BESSEL INTERPOLATION
C
      DIF=(DI-1.)/4.
      DJF=(DJ-1.)/4.
      A1=A(IN,JN+2)+DI*(A(IN+1,JN+2)-A(IN,JN+2)+DIF*
     +   (A(IN+2,JN+2)-A(IN+1,JN+2)+A(IN-1,JN+2)-A(IN,JN+2)))
      A2=A(IN,JN+1)+DI*(A(IN+1,JN+1)-A(IN,JN+1)+DIF*
     +   (A(IN+2,JN+1)-A(IN+1,JN+1)+A(IN-1,JN+1)-A(IN,JN+1)))
      A3=A(IN,JN)+DI*(A(IN+1,JN)-A(IN,JN)+DIF*
     +   (A(IN+2,JN)-A(IN+1,JN)+A(IN-1,JN)-A(IN,JN)))
      A4=A(IN,JN-1)+DI*(A(IN+1,JN-1)-A(IN,JN-1)+DIF*
     +   (A(IN+2,JN-1)-A(IN+1,JN-1)+A(IN-1,JN-1)-A(IN,JN-1)))
      B(I,J)=A3+DJ*(A2-A3+DJF*(A1-A2+A4-A3))
      GO TO 100
C
C     4-POINT LINEAR INTERPOLATION
C
  200 A1=DI*A(IN+1,JN+1)+(1.-DI)*A(IN,JN+1)
      A2=DI*A(IN+1,JN)+(1.-DI)*A(IN,JN)
      B(I,J)=DJ*A1+(1.-DJ)*A2
  100 CONTINUE
      CALL FILT25(B,IPS,JPS,C,1,IPS,1,JPS)
      CALL DMSPUT(DFILE,KEYP,B,ISTATP)
      IF(ISTATP .NE. 0) THEN
         PRINT 5040, KEYP
 5040    FORMAT(1X,'DATA FIELD "',A24,'" WRITEN NOT SUCCESSFULLY')
      END IF
      IF(KEYP(2:3) .EQ. '21') CALL WINDCV(B,C,IPS,JPS)
```

```
      IF(IAB .EQ. 1) THEN
         IAB=IAB+1
         GO TO 2222
      ELSE
         GO TO 1111
      END IF
 9999 CALL DMSCLS(DFILE,ISTAT)
      STOP 'PROGRAM "LTLNIJ" DONE !!!'
      END
C
C-----------------------------------------------------------------------
C
      SUBROUTINE FILT25(ZI,ILN,ILT,Z,LIS,LI,LJS,LJ)
C
C     SUBROUTINE TO PERFORM DATA SMOOTHING
C
      DIMENSION Z(ILN,ILT),ZI(ILN,ILT)
      LI1=LI-1
      LJ1=LJ-1
      LI2=LI1-1
      LJ2=LJ1-1
      LIS1=LIS+1
      LJS1=LJS+1
      LIS2=LIS1+1
      LJS2=LJS1+1
C
C     SMOOTH BOUNDARIES WITH 3-POINT (1-2-1) SMOOTHER
C
      DO 100 J=LJS1,LJ1
      Z(LIS,J)=(ZI(LIS,J+1)+ZI(LIS,J-1)+2.*ZI(LIS,J))/4.
  100 Z(LI,J)=(ZI(LI,J+1)+ZI(LI,J-1)+2.*ZI(LI,J))/4.
      DO 120 I=LIS1,LI1
      Z(I,LJS)=(ZI(I+1,LJS)+ZI(I-1,LJS)+2.*ZI(I,LJS))/4.
  120 Z(I,LJ)=(ZI(I+1,LJ)+ZI(I-1,LJ)+2.*ZI(I,LJ))/4.
C
C     SMOOTH FIRST INTERIOR POINTS WITH 9-POINT (1-2-1) SMOOTHER
C
      DO 185 J=LJS1,LJ1
      Z(LIS1,J)=(ZI(LIS2,J+1)+ZI(LIS,J-1)+ZI(LIS2,J-1)+ZI(LIS,J+1)
     +         +2.*(ZI(LIS2,J)+ZI(LIS1,J+1)+ZI(LIS1,J-1)+ZI(LIS,J))
     +         +4.*ZI(LIS1,J))/16.0
  185 Z(LI1,J) =(ZI(LI,J+1)+ZI(LI2,J-1)+ZI(LI,J-1)+ZI(LI2,J+1)
     +         +2.*(ZI(LI,J)+ZI(LI1,J+1)+ZI(LI1,J-1)+ZI(LI2,J))
     +         +4.*ZI(LI1,J))/16.0
      DO 195 I=LIS1,LI1
      Z(I,LJS1)=(ZI(I+1,LJS2)+ZI(I-1,LJS)+ZI(I+1,LJS)+ZI(I-1,LJS2)
     +         +2.*(ZI(I+1,LJS1)+ZI(I,LJS2)+ZI(I,LJS)+ZI(I-1,LJS1))
     +         +4.*ZI(I,LJS1))/16.0
  195 Z(I,LJ1) =(ZI(I+1,LJ)+ZI(I-1,LJ2)+ZI(I+1,LJ2)+ZI(I-1,LJ)
     +         +2.*(ZI(I+1,LJ1)+ZI(I,LJ)+ZI(I,LJ2)+ZI(I-1,LJ1))
     +         +4.*ZI(I,LJ1))/16.0
C
C     CORNER POINTS
C
      Z(LIS,LJS)=(Z(LIS1,LJS)+Z(LIS,LJS1)+Z(LIS1,LJS1))/3.0
      Z(LIS,LJ) =(Z(LIS,LJ1) +Z(LIS1,LJ) +Z(LIS1,LJ1))/3.0
      Z(LI,LJS) =(Z(LI1,LJS) +Z(LI,LJS1) +Z(LI1,LJS1))/3.0
      Z(LI,LJ)  =(Z(LI1,LJ)  +Z(LI,LJ1)  +Z(LI1,LJ1))/3.0
```

```
C
C     FILTER ALL POINTS WITHIN FIRST INTERIOR BOUNDARY WITH
C     BLECK'S 25-POINT FILTER.
C
      DO 220 I=LIS2,LI2
      DO 220 J=LJS2,LJ2
  220 Z(I,J)=0.279372*ZI(I,J)+0.171943*(ZI(I-1,J)+ZI(I,J-1)+ZI(I+1,J)
     +      +ZI(I,J+1))-0.006918*(ZI(I-2,J)+ZI(I,J-2)+ZI(I+2,J)
     +      +ZI(I,J+2))+0.077458*(ZI(I-1,J-1)+ZI(I+1,J+1)
     +      +ZI(I+1,J-1)+ZI(I-1,J+1))
     +      -0.024693*(ZI(I-1,J-2)+ZI(I+1,J-2)+ZI(I-2,J-1)
     +      +ZI(I+2,J-1)+ZI(I-2,J+1)+ZI(I+2,J+1)+ZI(I-1,J+2)
     +      +ZI(I+1,J+2))-0.01294*(ZI(I-2,J-2)+ZI(I+2,J-2)+ZI(I-2,J+2)
     +      +ZI(I+2,J+2))
      DO 240 I=LIS,LI
      DO 240 J=LJS,LJ
  240 ZI(I,J)=Z(I,J)
      RETURN
      END
C
C-----------------------------------------------------------------------
C
      SUBROUTINE PSGRID(AI,AJ,ALAT,ALON,IJLTN,OI,OJ,TLAT,DX,DY,AXSLON)
C
C     SUBROUTINE TO PERFORM TRANSFORMATION BETWEEN A RECTANGULAR GRID
C     POINT ON A POLAR STEREOGRAPHIC PROJECTION MAP AND LATITUDE/
C     LONGITUDE ON THE EARTH'S SURFACE BY GIVING GRID INDICES OF THE
C     POLE AND THE LONGITUDE OF THE AXIS ON THE PROJECTION MAP WHICH
C     IS PARALLEL TO THE X-AXIS.
C
C     AI......I-INDEX OF THE RECTANGULAR GRID POINT WHICH IS TO BE
C             TRANSFORMED
C     AJ......J-INDEX OF THE RECTANGULAR GRID POINT WHICH IS TO BE
C             TRANSFORMED
C     ALAT....RETURNED LATITUDE ON THE EARTH'S SURFACE FOR THE
C             RECTANGULAR GRID POINT (AI,AJ)
C     ALON....RETURNED LONGITUDE ON THE EARTH'S SURFACE FOR THE
C             RECTANGULAR GRID POINT (AI,AJ)
C     IJLTN...INDEX WHICH INDICATES DIRECTION OF TRANSFORMATION
C             = 0 TRANSFORM FROM (AI,AJ) TO (ALAT,ALON)
C             = 1 TRANSFORM FROM (ALAT,ALON) TO (AI,AJ)
C     OI......I-INDEX OF THE POLE
C     OJ......J-INDEX OF THE POLE
C     TLAT....TRUE LATITUDE FOR POLAR STEREOGRAPHIC PROJECTION
C     DX......GRID SPACING IN UNITS OF KM IN THE X-DIRECTION
C     DY......GRID SPACING IN UNITS OF KM IN THE Y-DIRECTION
C     AXSLON..LONGITUDE WHERE X-AXIS OF RECTANGULAR GRID
C             PARALLEL TO MAP PROJECTION
C
C     CONVENTION FOR LATITUDE AND LONGITUDE IS THAT POSITIVE FOR
C     EASTERN AND NORTHERN HEMISPHERES AND NEGATIVE FOR WESTERN
C     AND SOUTHERN HEMISPHERES.  ORIGIN OF X,Y-COORDINATES IS AT
C     THE POLE, X-COORDINATE INCREASES TO THE RIGHT AND Y-COORDINATE
C     INCREASES TO THE TOP OF MAP.  I-INDEX INCREASES TO THE RIGHT
C     AND J-INDEX INCREASES TO THE TOP WITH POINT (1,1) AT THE
C     BOTTOM-LEFT CORNER.
C
C     POLAR STEREOGRAPHIC PROJECTION HAS THE SAME DEFINITION AS THAT
```

```
C     OF LAMBERT CONFORMAL PROJECTION IF SET AN=1 AND TLAT1=TLAT2
C     IN PAGE 19 OF HOKE ET AL., 1981.
C
C     REFERENCES:
C     HOKE, J.E., J.L. HAYES AND L.G. RENNINGER, 1981: MAP PROJECTIONS
C          AND GRID SYSTEMS FOR METEOROLOGICAL APPLICATIONS.  U.S. AIR
C          FORCE GLOBAL WEATHER CENTER, AFGWC/TN-79/003, 86PP.
C     JENNE, R.L., 1970: THE NMC OCTAGONAL GRID.  NCAR MEMO.
C
      COMMON /ANG/ PI4, PI, DEGRAD
      A=6371.2213
      H=TLAT/ABS(TLAT)
      T=TLAT*DEGRAD
      XO=0.
      YO=0.
C
C     TRANSFORMATION FROM GRID INDICES TO LATITUDE/LONGITUDE
C
      IF(IJLTN .NE. 0) GO TO 1000
      XA=XO+(AI-OI)*DX
      YA=YO+(AJ-OJ)*DY*H
      IF(XA .EQ. 0.) THEN
         IF(YA .GT. 0.) TA= PI/2.
         IF(YA .EQ. 0.) TA= 0.
         IF(YA .LT. 0.) TA=-PI/2.
         RA=ABS(YA)
      ELSE
         TA=ATAN(YA/XA)
         IF(XA .LT. 0.) TA=TA-PI
         RA=XA/COS(TA)
      END IF
      ALON=AXSLON+TA/DEGRAD
      IF(ALON .GT.  180.) ALON=ALON-360.
      IF(ALON .LT. -180.) ALON=ALON+360.
      ALAT2=PI4*H-ATAN(TAN(PI4*H-T/2.)*(RA/(A*COS(T))))
      ALAT=2.*ALAT2/DEGRAD
      IF(ALAT .GT. 90.) ALAT=180.-ALAT
      RETURN
C
C     TRANSFORMATION FROM LATITUDE/LONGITUDE TO GRID INDICES
C
 1000 RA=A*COS(T)*(TAN(PI4*H-ALAT*DEGRAD/2.)/TAN(PI4*H-T/2.))
      TA=(ALON-AXSLON)*DEGRAD
      XA=RA*COS(TA)
      YA=RA*SIN(TA)*H
      AI=OI+(XA-XO)/DX
      AJ=OJ+(YA-YO)/DY
      RETURN
      END
C
C-----------------------------------------------------------------------
C
      SUBROUTINE UVCVRT(IG,JG,UT,VT,UG,VG,OI,OJ,ITG)
C
C     SUBROUTINE TO PERFORM CONVERSION OF WIND U,V-COMPONENTS BETWEEN
C     RECTANGULAR POLAR STEREOGRAPHIC GRID COORDINATES AND TRUE E-W/N-S
C     COORDINATES.
C
```

```
C      IG......I-COORDINATE OF THE GRID POINT
C      JG......J-COORDINATE OF THE GRID POINT
C      UT......U-COMPONENT ON TRUE COORDINATES
C      VT......V-COMPONENT ON TRUE COORDINATES
C      UG......U-COMPONENT ON GRID COORDINATES
C      VG......V-COMPONENT ON GRID COORDINATES
C      OI......I-COORDINATE OF THE ORIGIN OF RECTANGULAR GRID
C      OJ......J-COORDINATE OF THE ORIGIN OF RECTANGULAR GRID
C      ITG.....INDEX INDICATING CONVERSION DIRECTION
C              = 0 FROM TRUE COORDINATES TO GRID COORDINATES
C              = 1 FROM GRID COORDINATES TO TRUE COORDINATES
C
       CHARACTER*7  DFILE
       CHARACTER*24 KEYG, KEYP
       COMMON /DMS/ DFILE, KEYG, KEYP
       COMMON /ANG/ PI4, PI, DEGRAD
       XI=FLOAT(IG)-OI
       YJ=FLOAT(JG)-OJ
C
C      REVERSE DIRECTIONS OF WIND COMPONENTS FOR SOUTHERN HEMISPHERE
C
       IF(KEYP(7:7) .EQ. 'S') THEN
          UT=-UT
          VT=-VT
       END IF
C
C      THETA = ANGLE BETWEEN THE GRID X-COORDINATE AND GRID POSITION
C              VECTOR, POSITIVE COUNTER-CLOCKWISE
C      ALPHA = ANGLE BETWEEN TRUE AND GRID COORDINATES
C
       IF(XI .EQ. 0.) THEN
          IF(YJ .GT. 0.) ALPHA=PI/2.+PI/2.
          IF(YJ .EQ. 0.) ALPHA=0.
          IF(YJ .LT. 0.) ALPHA=PI/2.-PI/2.
       ELSE
          THETA=ATAN(YJ/XI)
          IF(XI .LT. 0.) THETA=THETA-PI
          ALPHA=PI/2.+THETA
       END IF
       SINA=SIN(ALPHA)
       COSA=COS(ALPHA)
       IF(ITG .EQ. 0) THEN
          UG= UT*COSA-VT*SINA
          VG= UT*SINA+VT*COSA
       ELSE
          UT= UG*COSA+VG*SINA
          VT=-UG*SINA+VG*COSA
       END IF
       RETURN
       END
C
C---------------------------------------------------------------------
C
       SUBROUTINE WINDCV(U,V,IPS,JPS)
C
C      SUBROUTINE TO PERFORM CONVERSION OF WIND U,V-COMPONENTS FROM
C      TRUE E-W/N-S COORDINATES TO RECTANGULAR POLAR STEREOGRAPHIC
C      GRID COORDINATES.
```

```
C
      DIMENSION U(IPS,JPS), V(IPS,JPS)
      CHARACTER*7  DFILE
      CHARACTER*24 KEYG, KEYP
      CHARACTER*24 KEYU, KEYV
      COMMON /DMS/ DFILE, KEYG, KEYP
C
C     KEYU = DMS RECORD KEY FOR INPUT U-COMPONENT
C     KEYV = DMS RECORD KEY FOR INPUT V-COMPONENT
C
      KEYU(1:24)=KEYP(1:24)
      KEYU(2:3)='20'
      KEYV(1:24)=KEYP(1:24)
      KEYV(2:3)='21'
      CALL DMSGET(DFILE,KEYU,U,ISTATU)
      IF(ISTATU .NE. 0) THEN
         PRINT 5000, KEYU
 5000    FORMAT(1X,'READ ERROR IN WIND COMPONENT "',A24,'"'/
     +          1X,'NO COORDINATES TRANSFORMATION DONE !!!')
         RETURN
      END IF
      CALL DMSGET(DFILE,KEYV,V,ISTATV)
      IF(ISTATV .NE. 0) THEN
         PRINT 5000, KEYV
         RETURN
      END IF
      DO 100 I=1,IPS
      DO 100 J=1,JPS
      UA=U(I,J)
      VA=V(I,J)
  100 CALL UVCVRT(I,J,UA,VA,U(I,J),V(I,J),32.,32.,0)
      CALL DMSPUT(DFILE,KEYU,U,ISTATU)
      IF(ISTATU .NE. 0) THEN
         PRINT 5050, KEYU
 5050    FORMAT(1X,'WARNING !!!  WRITE ERROR IN WIND FIELD "',A24,'"')
      END IF
      CALL DMSPUT(DFILE,KEYV,V,ISTATV)
      IF(ISTATV .NE. 0) PRINT 5050, KEYV
      RETURN
      END
```

附錄四

```
DMS file name:    MASCWB
Residence:        FE CYBER-840
Purpose:          A master file for gridded fields of NWP products
Format:           Floating point value or integer depending upon
                  record type specification in the record key
                  (grid orientation is defined by grid ID and
                   is described in the GRIDDEF file)
Key defination:   24 characters defined as the following,
                  1 - 3: field identifier,
                  4 - 6: TAU (forecast period, zero filled),
                  7 - 7: grid ID (G-144x73 global lat/lon grid,
                                  N-63x63 northern hemispheric
                                    grid on polar stereographic
                                    projection map,
                                  S-63x63 southern hemispheric
                                    grid on polar stereographic
                                    projection map,
                                  R-111x81 regional grid on Lambert
                                    conformal projection map,
                                  M-121x106 mesoscale grid on
                                    Lambert conformal projection map,
                                  T-131x111 typhoon grid on Lambert
                                    conformal projection map),
                  8 - 8: 'A' for objective analysis or forecast fields
                  9 -16: date-time group YYMMDDHH,
                     17: record type (F- floating point,
                                      I- integer,
                                      C- character     ),
                 18 -24: record length in units of record type
                              (zero filled)

Field identifier (3 characters) in the record key:
      (a) 3-D fields from analysis and forecast models:
          First character defines vertical pressure (or sigma) levels,
                A : sea-level surface (including land and sea)
                B : ocean surface
                C : 1000 mb
                D : 850 mb
                E : 700 mb
                F : 500 mb
                G : 400 mb
                H : 300 mb
                I : 200 mb
                J : 150 mb
                K : 100 mb
                L :  50 mb
                M :  30 mb
                N :  10 mb
                R : 925 mb
                T : 250 mb
          Second and third characters define the variable type
                00: geopotential height (input d-value)      (m)
                01: pressure                                 (mb)
                06: geopotential height anomaly              (m)
                10: temperature                              (degree C)
```

12:	vapor pressure	(mb)
13:	dew point depression	(degree C)
15:	relative humidity	(%)
20:	u-component of wind	(m/s)
21:	v-component of wind	(m/s)
22:	wind speed	(knots)
23:	wind direction	(degree)
30:	divergence	(1/s)
31:	vorticity	(1/s)
32:	vertical velocity	(10**-3 mb/s)
91:	streamline (need u/v-components)	
92:	wind vector (need u/v-components)	
93:	wind barb (need u/v-components)	
94:	wind vector/isotach (need u/v-components)	
95:	wind barb/height (need u/v-components and height)	

(b) single level fields:

A01:	sea-level pressure	(mb)
A04:	surface pressure tendency	(mb/day)
A08:	ground temperature	(degree C)
A10:	surface air temperature	(degree C)
A60:	total surface heat flux	(watts/m2)
A61:	surface longwave radiative heat flux	(watts/m2)
A62:	surface solar radiative heat flux	(watts/m2)
A63:	surface sensible heat flux	(watts/m2)
A64:	surface latent heat (moisture) flux	(watts/m2)
A65:	total precipitation rate	(mm/day)
A66:	convective precipitation rate	(mm/day)
A67:	large-scale precipitation rate	(mm/day)

Objective Analysis on a Polar Stereographic Projection Grid System Suitable for Use in the Global Forecast Model

Song-Chin Lin*, Feng-Lun Chiao**, Lie-Shieu Fang**

* Institute of Atmospheric Physics, National Central University
** Central Weather Bureau

ABSTRACT

For the CWB global forecast model, the computations are basically conducted on a latitude/longitude grid system. However, it would be impractical to display the forecast results on a rectangular global or hemispheric map. The purpose of this study is to establish a set of coordinates transformation formulae and computer programs which will enable the display of forecast results on a polar stereographic projection map.